

Generating and Exploring S-Box Multivariate Quadratic Equation Systems with SageMath

A.-M. Leventi-Peetz¹ and J.-V. Peetz²

¹ Federal Office for Information Security,
Godesberger Allee 185–189, DE-53175 Bonn, Germany
leventi-peetz@bsi.bund.de

² DE-53343 Wachtberg, Germany
jvpeetz@web.de

May, 2017

Abstract

A new method to derive Multivariate Quadratic equation systems (MQ) for the input and output bit variables of a cryptographic S-box from its algebraic expressions with the aid of the computer mathematics software system SageMath is presented. We consolidate the deficiency of previously presented MQ metrics, supposed to quantify the resistance of S-boxes against algebraic attacks.

Key Words – Algebraic attack resistance, algebraic cryptanalysis, Lagrange polynomial, multivariate quadratic polynomial equation system, polynomial quotient ring, SageMath, SAT solver, S-box, Rijndael AES.

1 Overview

We present a new automated way to produce and investigate Multivariate Quadratic equation systems (MQ) over $GF(2)$ for the Rijndael S-box (S_{RD}) and alike. In the next we first shortly survey the principles of S_{RD} in section 2. Recently, Jie Cui *et al.* [1] claimed to have presented a new and concise approach for generating such MQ for S_{RD} and also to have proposed a cryptographically more secure S-box. In section 3 we depict the derivation of Cui *et al.* In section 4 we present our automated way to produce the MQ for S_{RD}

aided by the computer mathematics software system SageMath [2]. As we demonstrate later, this method can be applied for other proposed S-boxes, alleged to be cryptographically stronger. We also generate the Gröbner bases describing the two different S-boxes and calculate solutions of all equation systems with the help of a SAT solver. In section 5 we investigate, critically discuss, and dismiss equation-metrics-based criteria as inappropriate to estimate the resistance against algebraic attacks (RAA) of an S-box. In sections 6 and 7 we build MQ for the S-box proposed by Jie Cui *et al.* [1] and show why this is not an improvement. The complicated algebraic expression for the S-box constructed by Cui *et al.* leads to a great number of equations and independent monomials in the resulting MQ. On the basis of the RAA formulas the new S-box should demonstrate a remarkable robustness against algebraic attacks. But from that same S-box can be derived a much simpler MQ leading to a polynomial system nearly as easy solvable as that of the original S_{RD} . This is what we do in section 7 greatly facilitated by SageMath.

Our main contribution is the demonstration of the handiness which SageMath offers to the researcher so that he can derive his MQ and its metrics in a fast way and transparently verify the quality of existing formulas supposed to quantify resistance of the MQ to algebraic attacks. In conclusion, we couldn't validate the predicted reduced hardness for a system to solve with increasing number of equations and number of independent monomials in the polynomial systems according to suggested RAA formulas.

2 Principle of Rijndael S-box RD

We shortly repeat its well known principles and algebraic properties [3]. Looking upon 8-bit bytes as elements in $GF(2^8)$, Rijndael's S-box is a mapping $S : GF(2^8) \rightarrow GF(2^8)$ in form of a combination of an inverse function $I(\mathbf{x})$ which is a multivariate inverse modulo the irreducible polynomial $m(t) = t^8 + t^4 + t^3 + t + 1$ and an affine transformation function $A(\mathbf{x})$. \mathbf{x} is a byte variable consisting of bits $x_i (i = 0, \dots, 7)$, with x_7 symbolizing the most significant bit: $\mathbf{x} = \sum_{i=0}^7 x_i t^i$. The modular inverse function $I(\mathbf{x})$ is defined as:

$$I(\mathbf{x}) = \mathbf{x}^{254} \bmod m(t) \quad (1)$$

i.e., the modular inverse of 0 is mapped to 0. According to the AES design [4, 5], the affine transformation $A(\mathbf{x})$ can also be described as a modular polynomial multiplication followed by an addition (XOR) of a constant polynomial:

$$A(\mathbf{x}) = \mathbf{a} \mathbf{x} \bmod (t^8 + 1) + \mathbf{b} \quad (2)$$

with $\mathbf{a} = \text{'1F'}$ and $\mathbf{b} = \text{'63'}$. A two-digit hexadecimal number stands for a *constant* byte, that is a polynomial in t , e.g, '63' for $t^6 + t^5 + t + 1$. The Rijndael S-box can be written as:

$$S_{\text{RD}}(\mathbf{x}) = A \circ I = A(I(\mathbf{x})) \quad (3)$$

3 Rijndael S-box explored by Cui et al.

Cui *et al.* [1] (as Courtois and Pieprzyk [3] before them) utilize the Rijndael S-box composition (3) to derive an MQ for it. With \mathbf{x} the input and \mathbf{z} the output value, and the intermediate variable $\mathbf{y} = I(\mathbf{x})$ they note: $\mathbf{z} = S_{\text{RD}}(\mathbf{x}) = A(\mathbf{y}) = A(I(\mathbf{x}))$. Considering the inverse transformation $\mathbf{y} = I(\mathbf{x})$, obviously $\mathbf{x}\mathbf{y} = 1$ when \mathbf{x} not equal 0, which reads in polynomial form:

$$\left(\sum_{i=0}^7 x_i t^i \right) \left(\sum_{j=0}^7 y_j t^j \right) \bmod m(t) = 1 \quad (4)$$

The above modulo division is then analytically performed and a comparison of coefficients of terms of the same order in t^k , $0 \leq k \leq 7$, leads to the first eight multivariate quadratic equations for Rijndael S-box on the pages 2483, 2484 of the paper of Cui *et al.* [1] The authors give all the steps and in-between results of the complete length of the calculation. They formulate and evaluate two additional equations of the byte variables to define the S-box completely. Doing so, Cui *et al.* replicate results already presented in 2002 by Courtois and Pieprzyk in the extended version of [3].

4 Rijndael S-box coded in Sage

SageMath or briefly Sage (System for Algebra and Geometry Experimentation) is a free open-source software system for computer mathematics [6]. It is licensed under the Gnu General Public License. It builds on top of many existing computer mathematics open-source packages. Their combined power is accessible through a common, Python-based language interface from the command line or a web browser. Originally, it is designed by William Stein, still the leader of the SageMath project, and also inventor of SageMath-Cloud [7] for collaborative computational mathematics.

In order to work with polynomials like $\mathbf{x} = \sum_{i=0}^7 x_i t^i$, Sage provides modules to construct rings of multivariate polynomials. The polynomials \mathbf{x} , \mathbf{y} , and \mathbf{z} introduced in the previous section we model in Sage as follows.¹ In

¹The complete code presented here together with its output is accessible at SageMath-Cloud [8].

line 1 of listing 1 a variable for the number of bits is defined for convenience. In line 2 a list of strings for the names of the coefficients of the three byte polynomials is generated (`['x0', 'x1', ..., 'z7']`).

Listing 1: Byte polynomials over a quotient ring

```

1  nb = 8
2  var1 = [c + str(p) for c in 'xyz' for p in range(nb)]
3  B = BooleanPolynomialRing(names = var1)
4  B.inject_variables()
5  P.<p> = PolynomialRing(B)
6  Byte.<t> = P.quotient_ring(p^8 + p^4 + p^3 + p + 1)
7  X = B.gens()[ :nb]
8  Y = B.gens()[nb:2*nb]
9  x = sum([X[j]*t^j for j in range(nb)])
10 y = Byte(list(Y))

```

In line 3 a Boolean polynomial ring for these coefficients is constructed which assigns $GF(2)$ properties to them. In line 4 the coefficient names are made available as variables. In line 5 a polynomial ring over the Boolean polynomial ring B is constructed and from that, in line 6, the final quotient ring Byte with modulus $m(t)$. In lines 7 and 8, lists² of coefficient variables of the byte polynomials are created for convenience. With the help of these lists, in the last two lines the polynomials are modeled in two equivalent ways, \mathbf{x} explicitly, and \mathbf{y} by using the `Byte` constructor.

Now one can already evaluate the product \mathbf{xy} in Sage with the commands:

```

E3 = x * y
eqs3 = E3.list()

```

In the second line we used the `list()` attribute to get the coefficients of each power of t in expression $\mathbf{E3}$. Due to the usage of the quotient ring, $\mathbf{E3}$ is of degree 7, the length of list `eqs3` (the number of coefficients) is 8. The terms we have gotten with Sage compare with the right-hand sides (rhs) of the system of equations with number (3) in the paper of Cui *et al.* [1]

Cui *et al.* proceeded with the generation of the next set of equations for Rijndael's S-box, the affine transformation. From equation (2) setting $\mathbf{z} = A(\mathbf{y})$ it follows

$$\mathbf{y} = \mathbf{a}^7(\mathbf{z} + \mathbf{b}) \bmod(t^8 + 1) \quad (5)$$

since $\mathbf{a}^7 \mathbf{a} \bmod(t^8 + 1) = 1$. Substituting (5) for \mathbf{y} in \mathbf{xy} we get the final form of the first implicit eight equations representing Rijndael S-box. In Sage the values substitution is accomplished with the help of a so called *dictionary*.

²To be exact, in Python these are *tuples*, i.e., immutable *lists*.

By using equation (5) the code in Listing 2 generates this dictionary, called eqs4.

Listing 2: Generate a dictionary to substitute \mathbf{y} variables

```
Baff.<u> = P.quotient_ring(p^8 + 1)
Z = B.gens()[2*nb:][:nb]
z = Baff(list(Z))
a = u^4 + u^3 + u^2 + u + 1
b = u^6 + u^5 + u + 1
eqs4 = dict(zip(Y, (a^7 * (z + b)).list()))
```

The first line sets up a quotient ring modulo $t^8 + 1$. The next four lines define the byte variable \mathbf{z} and the two constant polynomials \mathbf{a} and \mathbf{b} in this ring (with generator u). The rhs of equation (5) simply reads $\mathbf{a}^7 * (\mathbf{z} + \mathbf{b})$ in the code. The dictionary is constructed in the last line. The result is shown in Listing 3.

Listing 3: Dictionary to substitute \mathbf{y} variables

```
{y7: z6 + z4 + z1,
 y6: z5 + z3 + z0,
 y5: z7 + z4 + z2,
 y4: z6 + z3 + z1,
 y3: z5 + z2 + z0,
 y2: z7 + z4 + z1 + 1,
 y1: z6 + z3 + z0,
 y0: z7 + z5 + z2 + 1}
```

The substitution of \mathbf{y} in equation eqs3 via the dictionary succeeds with the following:

```
eqs5 = [_ .subs(eqs4) for _ in eqs3]
```

The result is again a list, the members of which give the first set of eight multivariate quadratic equations of the S-box by setting the byte variable product equal to 1. This list of terms eqs5 is identical to the system of equations (5) of Cui *et al.* Those equations with zero constant term (7 out of 8 above) are true with probability equal to 1. The 8th equation (the coefficient of t^0) is true only when $\mathbf{x} \neq 0$, so that this equation is true with a probability 255/256. Furthermore for $\forall \mathbf{x} \neq 0 \mathbf{x} = \mathbf{x}^2 \mathbf{y}$. Obviously this last equation is true also when $\mathbf{x} = 0$, so that one can write:

$$\forall \mathbf{x} \in GF(2^8) \begin{cases} \mathbf{x} & = \mathbf{y} \mathbf{x}^2 \\ \mathbf{x}^2 & = \mathbf{y}^2 \mathbf{x}^4 \\ & \vdots \\ \mathbf{x}^{128} & = \mathbf{y}^{128} \mathbf{x}^{256} = \mathbf{y}^{128} \mathbf{x} \end{cases} \quad (6)$$

Cui *et al.* take the last of the above and write two symmetrical equations to generate an additional set of 16 equations for the Rijndael S-box. The equations they take are:

$$\begin{cases} \mathbf{x}^{128} = \mathbf{y}^{128}\mathbf{x} \\ \mathbf{y}^{128} = \mathbf{x}^{128}\mathbf{y} \end{cases} \quad (7)$$

We develop the two last equations to get the needed additional 16 equations for the implicated variables. We also substitute in these equations \mathbf{y} by using the dictionary in listing 3.

```
E7 = x^128 + y^128 * x
eqs7 = [_ .subs(eqs4) for _ in E7.list()]
```

The result is a list of terms which are practically the equations (7) of Cui *et al.* written in the reverse order than that of Cui's paper. Cui *et al.* begin with the expression corresponding to the highest order term of $E7$ while the sage list begins with the constant term. These terms are identical with the rhs of equations (7) in Cui *et al.* [1] with a couple of minimal differences which we attribute to typographical errors in the paper of Cui *et al.*

Similarly, we write:

```
E8 = y^128 + x^128 * y
eqs8 = [_ .subs(eqs4) for _ in E8.list()]
```

and, by setting these terms equal to 0, get the next block of eight equations for the Rijndael S-box which are to be compared with the system (8) of Cui *et al.* Here we see a couple of discrepancies which we again attribute to typographical mistakes in the reference paper.

Using the Sage model of this S-box it is easy to count the number of equations, the number of terms in each equation and determining the minimal and maximal number of terms, as well as the total number of different terms, as shown in Listing 4.

Listing 4: Survey of first S-box equation system

```
mq1 = eqs5[1:] + eqs7 + eqs8
len(mq1)
lmon1 = [len(_ .monomials()) for _ in mq1]
min(lmon1)
max(lmon1)
Sequence(mq1).nmonomials()
```

As mentioned above, the first equation is discarded since it is only true with probability 255/256 (false if $\mathbf{x} = 0$). This gives for the Rijndael S-box 23 equations, with between 28 and 49 monomials per equation and, in total, 81 different monomials.

Finding the 256 solutions of this equation system representing the value table of the byte S-box with a SAT solver in Sage is accomplished with the following two lines of code:

Listing 5: SAT solver usage

```
from sage.sat.boolean_polynomials \
    import solve as solve_sat
%time r = solve_sat(mq1, n=infinity)
```

This takes ca. 0.6 s CPU time on a decent computer (2.8 GHz CPU, 8 GB RAM). As a point of reference we also evaluate the Gröbner basis (GB) of this MQ and print the number of the basis equations, as well as the maximal degree and the number of its monomials:

Listing 6: Gröbner basis evaluation

```
Id11 = B.ideal(mq1)
%time mq1gb = Id11.groebner_basis()
print len(mq1gb)
print mq1gb.maximal_degree()
print mq1gb.nmonomials()
```

The evaluation of the Gröbner basis takes ca. 14 seconds, it has 8 equations of degree 7 with 263 different monomials. The solution of the basis equations with the SAT solver is about 4 times as fast as the solution of the MQ.

Courtois and Pieprzyk [3] state that these 23 equations are linearly independent. Nonetheless, the last 16 equations (7) only, i.e. $\text{mq2} = \text{eqs7} + \text{eqs8}$, are already sufficient to evaluate the Gröbner basis and to compute the S-box value table of 256 solutions with the SAT solver which takes ca. 0.7 s CPU time on the same computer. The 16 equations describing the Rijndael S-box have between 28 and 49 monomials per equation and, in total, 81 different monomials.

5 Algebraic attacks and S-box optimization

For quantifying the resistance against algebraic attacks for r equations in t terms over $GF(2^n)$ Cui *et al.* [1] have used the criterion of Cheon and Lee [9] which defines the Resistance against Algebraic Attacks (RAA) Γ as:

$$\Gamma = \left(\frac{t-r}{n} \right)^{\lceil (t-r)/n \rceil} \quad (8)$$

Courtois and Pieprzyk [3] use another criterion

$$\Gamma_{\text{CP}} = \left(\frac{t}{n} \right)^{\lceil t/r \rceil} \quad (9)$$

(note the brackets $\lceil \cdot \rceil$ in the exponent indicating the ceiling function).³ The value of these criteria should reflect the difficulty of solving multivariate equations. For the Rijndael S-box we counted 23 equations and, in total, 81 different monomials. Therefore, it has $\Gamma = (29/4)^8 \approx 2^{22.9}$ and $\Gamma_{\text{CP}} = (81/8)^4 \approx 2^{13.4}$. The 16 equations system has $\Gamma = (65/8)^9 \approx 2^{27.2}$ and $\Gamma_{\text{CP}} = (81/8)^6 \approx 2^{20.0}$. Compared with the relation of the computational effort of the SAT solver for the MQ for 23 and 16 equations respectively, the Γ values for the 16 equation system are exaggerated.

To generate a *harder* to solve equation system Cui *et al.* [1] have introduced a more complicated Rijndael S-box structure which they name Affine-Inverse-Affine (AIA) structure. This S-box will be explored in detail in the next two sections.

6 MQ of the AIA structure S-box in Sage

In Cui, Huang, *et al.* (2011) [10], a new Rijndael S-box structure named *Affine-Inverse-Affine* (AIA) is designed supposed to increase the algebraic complexity of said S-box. Questioning this claim, we considered it worthwhile to try and check their calculations and assertions.

A different affine transformation (2) with $\mathbf{a} = \text{'5B'}$ and $\mathbf{b} = \text{'5D'}$ is chosen. This transformation is applied before and after the inversion step:

$$S_{\text{AIA}}(\mathbf{x}) = A \circ I \circ A = A(I(A(\mathbf{x})))$$

Cui *et al.* [1] derive a multivariate quadratic equation system of S_{AIA} using the coefficients of the polynomial expression of the S-box. They write down the equation system with indices for rounds and input bytes for the AES algorithm (but never use them). The round indices will be omitted here as they don't matter in what follows. As before, by \mathbf{x} is denoted the input byte variable of the S-box function. Intermediate variables are denoted by $\mathbf{y}_0, \mathbf{y}_1, \dots, \mathbf{y}_{253}$ and the output variable by \mathbf{z} . According to the polynomial expression of the new AIA S-box, the S-box transformation can be described by the following quadratic equations over $\text{GF}(2^8)$:

$$\begin{cases} \mathbf{x} \mathbf{y}_0 = 1 \\ \mathbf{y}_m \mathbf{y}_0 = \mathbf{y}_{m+1}, \text{ for } 0 \leq m \leq 252 \text{ and } \mathbf{y}_{253} = \mathbf{x} \\ \mathbf{z} = g(\mathbf{y}_0, \mathbf{y}_1, \dots, \mathbf{y}_{252}, \mathbf{x}) \end{cases} \quad (10)$$

Cui *et al.* [1] define the function g by the polynomial expression of their S-box. We calculated the coefficients for the polynomial expression of S_{AIA} (its

³Parameter n could be interpreted as number of dependent variables, see section 6.

Lagrange polynomial) in Sage⁴ and tabulate them in Listing 8. Thereby, the function g reads

$$g(\mathbf{y}_0, \mathbf{y}_1, \dots, \mathbf{y}_{252}, \mathbf{x}) = \text{'FA'} + \text{'12'} \mathbf{x} + \text{'26'} \mathbf{y}_{252} + \dots + \text{'E5'} \mathbf{y}_2 + \text{'A9'} \mathbf{y}_1 + \text{'A6'} \mathbf{y}_0 \quad (11)$$

(The coefficients of Cui *et al.* [1] as listed in their Table 1 which in turn corresponds to Table 3 in Cui, Huang, *et al.* [10] are wrong and don't represent the polynomial expression of their S-box S_{AIA} , although, Cui, Huang, *et al.* list in their preceding Table 2, correctly, the output values of S_{AIA} .)

The equation system (10) can be modeled in Sage with the help of the preparatory code shown in Listing 7:

Listing 7: Preparation for equation system of AIA S-box

```

1  nb = 8
2  ny = 253
3  varlzx = [c + str(p) for c in 'xz' for p in range(nb)]
4  varly = ['y' + str(p) for p in range(nb*ny)]
5  B = BooleanPolynomialRing(names = varlzx + varly)
6  B.inject_variables()
7  P.<p> = PolynomialRing(B)
8  Byte.<t> = P.quotient_ring(p^8 + p^4 + p^3 + p + 1)
9  X = B.gens()[ :nb]
10 Z = B.gens()[nb:][ :nb]
11 YY = [B.gens()[ (2+m)*nb: ][ :nb] for m in range(ny)]
12 x = Byte(list(X))
13 z = Byte(list(Z))
14 yy = [Byte(list(_Y)) for _Y in YY]
```

In lines 3 and 4 lists of strings for the names of coefficients for the byte variables \mathbf{x} , \mathbf{z} , and $\mathbf{y}_0, \dots, \mathbf{y}_{252}$ are generated.

Then, as in Listing 1, in line 5 a Boolean polynomial ring for these coefficients is constructed, assigning $GF(2)$ properties to them, in line 6 the coefficients are made available as variables and in line 7 a polynomial ring over the Boolean polynomial ring B is constructed and from that in line 8, eventually, the quotient ring `Byte` with modulus $m(t)$.

In lines 9 to 11 tuples of the coefficient variables of the byte polynomials are created for convenience. For the \mathbf{y} -variables the coefficients are grouped byte-wise in sub-lists. In the last three lines, finally, the polynomials of the byte variables are defined using these tuples as arguments for the `Byte` constructor. For the \mathbf{y} -variables a list of polynomials is used.

⁴The Sage code comprises some twenty lines and is accessible at SageMathCloud [11].

In the next Sage code block (Listing 8), the coefficients of the polynomial expression of S_{AIA} are given in hexadecimal notation as a list of strings which is transformed to a list of the constant polynomials that enter g (11):

Listing 8: Generating constant polynomials of AIA S-box

```
sbt = [
'FA 12 26 E7 9A C7 DB 79 56 01 D3 59 52 ED 97 C9',
'47 46 FC 7C 5A 50 49 BF F4 F8 63 C8 82 1B EE 74',
'3B 5D F8 02 2D 64 1A 15 BA DB 59 FE FB D6 97 FF',
'AB 3F B4 09 32 77 AB 52 4D 96 D5 BB DE 30 DE 05',
'62 23 7C 69 66 75 9F E9 9B 60 88 2F D1 8F 09 F4',
'1E EF C4 48 0D A5 AE 7A 38 9B 71 F2 9F 44 B3 99',
'20 C5 13 12 19 C2 5F 5B AD FA D5 49 7B F8 16 07',
'B6 75 E9 B0 CA E8 83 C1 4E 75 C5 5E 91 07 86 BF',
'6F C2 25 35 D3 7F CC 0D AC 7A C9 EC D2 3F C3 21',
'7E A9 2A 6D A8 66 F8 7D D2 1B FE CD 58 64 25 DA',
'AE 49 2D 4F 0C 74 F2 42 4A 87 42 9B 83 50 F1 91',
'C1 02 4F 2A C9 19 37 59 D5 74 8D 0B 20 C5 AF 28',
'47 FB 09 87 10 6A 3B C8 8B 08 5B 8B 13 0E 73 7E',
'FA 45 85 18 D5 90 4E 71 E6 F2 BF EE 30 E9 99 54',
'30 63 8F 03 92 91 0C 43 09 66 E5 76 6A 93 87 E4',
'6C 6A 87 A1 CB 64 AA 5C FB 05 5A DE E5 A9 A6 00']
sbt = ' '.join(sbt).split()
sbp = [Byte(ZZ(_, 16).bits()) for _ in sbt]
```

In the last line of this code block each two-digit hexadecimal number in the table represented by a two character string is converted into a decimal number by the code fragment `ZZ(_, 16)`. Appending `.bits()` transforms it into a list of 0s and 1s, a big-endian binary representation of the hexadecimal number. Applying the `Byte` constructor gives the corresponding constant polynomial.

With these preparations the equation system (10), (11) of the AIA S-box (equation 9 in Cui *et al.* [1]) can be modeled in Sage as shown in Listing 9.

Listing 9: Equation system of AIA S-box in Sage

```
g = sbp[0] + sbp[1] * x \
    + sum(sbp[2+m] * yy[ny-1-m] for m in range(ny))
yy.append(x)
E9 = [x * yy[0] + 1]
E9.extend(yy[m] * yy[0] + yy[m+1] for m in range(ny))
E9.append(z + g)
mq3 = flatten([_.list() for _ in E9])[1:]
```

In the last line the first term (the t^0 -coefficient of $\mathbf{xy}_0 + 1$) is discarded since

it is only true with probability $255/256$ (false if $\mathbf{x} = 0$). Using this Sage model we evaluate some metrics of this MQ as before. This gives for the new AIA S-box 2,039 equations, with between 3 and 1,034 monomials per equation. These equations have in total 18,232 different monomials. In order to apply the criteria of section 5, the intermediate variables also were taken into account by interpreting the parameter n in the definitions (8) and (9) as the number of dependent variables. Hence, the number 8×254 is used as n , and not only 8. This results in $\Gamma = (16,193/2,032)^8 \approx 2^{24.0}$ and $\Gamma_{\text{CP}} = (18,232/2,032)^9 \approx 2^{28.5}$ as estimation for RAA.⁵

Also, the CPU time to evaluate all 256 solutions of this MQ with a SAT solver is ca. 7 s, which is 12 ($\approx 2^4$) times as long as for the original Rijndael S-box.

Cui *et al.* [1] count a totally different number of equations and terms based on the byte variables, not on their polynomial coefficients, which contradicts the scheme applied to the Rijndael S-box with which they compare, and therefore, is misleading.

Further, this method used by Cui *et al.* [1] to derive an MQ for their AIA S-box applies to any S-box using the coefficients of its polynomial expression. This illustrates that the resulting numbers of equations and terms are deceptive as criterion for the estimation of algebraic attack resistance and inapt to differentiate the quality of byte S-boxes. To substantiate this point, we have derived such an MQ for the original Rijndael S-box by using its polynomial expression in equation (10). The function g then reads

$$\begin{aligned}
 g_{\text{SRD}}(\mathbf{y}_0, \mathbf{y}_1, \dots, \mathbf{y}_{252}, \mathbf{x}) = & \\
 & \text{'63'} + \text{'8F'} \mathbf{y}_{127} + \text{'B5'} \mathbf{y}_{63} + \text{'01'} \mathbf{y}_{31} + \text{'F4'} \mathbf{y}_{15} + \text{'25'} \mathbf{y}_7 + \\
 & \text{'F9'} \mathbf{y}_3 + \text{'09'} \mathbf{y}_1 + \text{'05'} \mathbf{y}_0
 \end{aligned} \tag{12}$$

The same Sage code (Listings 7, 8, and 9) was used with an adapted table of the polynomial expression coefficients in accordance with equation (12). This MQ of the Rijndael S-box exhibits the same number of equations with the same number of different monomials as the MQ of S_{AIA} resulting in equally high, misleading Γ values. Also, the SAT solver needs the same 7 s CPU time to find the solutions of this MQ.

In contrast, we will show in the next section how to derive, aided by computer mathematics, a much simpler MQ for the AIA S-box which shows that its resistance against algebraic attacks according to the effort of a SAT solver not really exceeds that of the original Rijndael S-box. But the RAA criteria exaggerate the hardness of that simpler MQ.

⁵Formal application of $n = 8$ yields unlikely high values: $\Gamma \approx 2^{22,241}$ and $\Gamma_{\text{CP}} \approx 2^{100}$.

7 Concise MQ for the AIA S-box in Sage

Building on the Sage code presented so far we derive a much simpler MQ for the AIA S-box. Its resistance against algebraic attacks according to the RAA criterion should be greater than that of the original Rijndael S-box but over-estimates the time it takes to solve the system with a SAT solver.

For the inversion step we now use, temporarily, two intermediate byte variables \mathbf{y}_0 and \mathbf{y}_1 , named `yy[0]` and `yy[1]` in the Sage code. Their coefficients shall be y_0, \dots, y_7 and y_8, \dots, y_{15} , respectively. The three steps of the AIA S-box are

$$\mathbf{z} = A(\mathbf{y}_1), \mathbf{y}_1 = I(\mathbf{y}_0), \mathbf{y}_0 = A(\mathbf{x})$$

Beginning with the inversion step we first model

$$\begin{cases} \mathbf{y}_0^{128} &= \mathbf{y}_1^{128} \mathbf{y}_0 \\ \mathbf{y}_1^{128} &= \mathbf{y}_0^{128} \mathbf{y}_1 \\ \mathbf{y}_0^3 &= \mathbf{y}_0^4 \mathbf{y}_1 \\ \mathbf{y}_1^3 &= \mathbf{y}_1^4 \mathbf{y}_0 \end{cases} \quad (13)$$

The last two equations in (13) are the only other additional (fully quadratic) MQ (besides $\mathbf{y}_0 \mathbf{y}_1 = 1$) for the inversion as stated already by Courtois and Pieprzyk in the extended version of [3] (compare also Cheon and Lee [9]). These additional equations are necessary to completely define the S-box S_{AIA} . Without them the system is under-defined, as, for example, the solution with a SAT solver shows. In Sage the equations (13) read

$$\begin{aligned} \text{E10} &= \text{yy}[0]^{128} + \text{yy}[1]^{128} * \text{yy}[0] \\ \text{E11} &= \text{yy}[1]^{128} + \text{yy}[0]^{128} * \text{yy}[1] \\ \text{E12} &= \text{yy}[0]^3 + \text{yy}[0]^4 * \text{yy}[1] \\ \text{E13} &= \text{yy}[1]^3 + \text{yy}[1]^4 * \text{yy}[0] \end{aligned}$$

The linear transformations according to equation (2) are

$$\mathbf{y}_0 = \mathbf{a} \mathbf{x} \bmod(t^8 + 1) + \mathbf{b} \quad (14)$$

$$\mathbf{y}_1 = \mathbf{a}^7(\mathbf{z} + \mathbf{b}) \bmod(t^8 + 1) \quad (15)$$

with $\mathbf{a} = \text{'5B'}$ (hence $\mathbf{a}^8 \bmod(t^8 + 1) = 1$) and $\mathbf{b} = \text{'5D'}$. In Sage we formulate

Listing 10: Generate dictionary to substitute \mathbf{y} variables

```

1 Baff.<u> = P.quotient_ring(p^8 + 1)
2 a = Baff(ZZ('0x5B').bits())
3 b = Baff(ZZ('0x5D').bits())
4 eqs14 = dict(zip(YY[0], a * Baff(x) + b))
5 eqs14.update(zip(YY[1], a^7 * (Baff(z) + b)))

```

The right hand sides of equations (14) and (15) enter Listing 10 in the two last lines. Both their coefficients are inserted into the same dictionary eqs14. Applying this substitutions in Sage to get rid of the intermediate byte variables is straight forward

```
eqs10 = [_ .subs(eqs14) for _ in E10.list()]
eqs11 = [_ .subs(eqs14) for _ in E11.list()]
eqs12 = [_ .subs(eqs14) for _ in E12.list()]
eqs13 = [_ .subs(eqs14) for _ in E13.list()]
```

and gives already the final, concise MQ for the S-box S_{AIA} :

```
mq5 = eqs10 + eqs11 + eqs12 + eqs13
```

For this MQ counting the numbers of equations and monomials in Sage is done as before. It has 32 equations, with between 33 and 60 monomials per equation and 137 different monomials in total. This makes according to definitions (8) and (9) $\Gamma = (105/8)^{14} \approx 2^{52}$ and $\Gamma_{CP} = (137/8)^5 \approx 2^{20.5}$.

Nonetheless, the SAT solver takes ca. 0.8 s CPU time on the same computer (2.8 GHz CPU, 8 GB RAM) to find all and only 256 solutions for this MQ. The evaluation of its Gröbner basis takes ca. 16 s. The GB has 8 equations of degree 7 with 263 different monomials and its solution with a SAT solver is obtained as fast as that of the GB of the Rijndael S-box. Clearly, the values of the hardness criteria do not correlate with the effort of the SAT solver for this MQ.

For comparison and as an additional reference value for the RAA estimation, we have derived such an MQ with 32 equations for the original Rijndael S-box using the four equations (13) (replacing \mathbf{y}_0 by \mathbf{x} and \mathbf{y}_1 by \mathbf{y}) and its affine transformation (5) (with $\mathbf{a} = '1F'$, $\mathbf{b} = '63'$). This MQ has the same number of equations and the same number of different monomials, thus, the same values for the hardness criteria as that of S_{AIA} . The solution with the SAT solver of this MQ for the original Rijndael S-box takes the same CPU time as the solution of the 32 equation MQ of S_{AIA} , namely, ca. 0.8 s. This shows how Sage can easily be used to disprove the practicality of the hardness criteria.

8 Conclusion

SageMath is a very appropriate, powerful computer mathematics tool to analyze cryptographic problems formulated with byte variables as polynomials in a quotient ring. Sage draws its strength in this area mainly from the integration of the BRiAl, former PolyBoRi, library [2, 12].

We have used Sage to demonstrate how to produce various polynomial

Multivariate Quadratic equation systems (MQ) as well as their Gröbner basis for the Rijndael S-box S_{RD} and similar S-boxes in a simple and straightforward manner. Using the flexible structures and interface of Sage one can easily evaluate metrics of the resulting polynomial systems, like the number of different monomials in the system, the length of the equations, the frequency of the appearance of certain terms or variables in equations etc. With this facility we generated the necessary inputs for the application of estimations of the Resistance against Algebraic Attacks (RAA) proposed by Cui *et al.* [1] or by Courtois and Pieprzyk [3]. Parallely, we performed numerical experiments by solving the corresponding MQ with a SAT solver using the required computing time as a measure for the RAA.

Our results in this respect are revealing. We couldn't validate the predicted reduced hardness for a system to solve with increasing number of equations and number of independent monomials in the polynomial systems which both the formulas forecast. There is in fact a slight increased solver effort when one reduces from the 23 Rijndael S-box equations to the 16 but quantitatively this is badly represented in both formulas.

Cui *et al.* constructed a complicated algebraic expression for a new Rijndael S-box S_{AIA} (starting from its Lagrange polynomial expression with 255 coefficients) which necessarily leads to a great number of equations and independent monomials in the resulting MQ. On this basis they thought they have demonstrated a remarkable new S-box practically not possible to solve according to the here discussed and by us dismissed RAA formulas. However there are gaps in their concept arising from inconsistency in their comparison principle as well as the lack of thoroughness in the investigation of the properties of the new algebraic expression which we showed can be equivalently written in a much simpler form leading to a polynomial system nearly as easy solved as that of the original Rijndael S-box.

We also mapped the original Rijndael S-box with its 9 Lagrange coefficients on the AIA form of Cui *et al.* which gave us as result the same *huge* number of variables and multitude of polynomials which should manifest that this is no way to create especially hard cryptographic S-boxes.

We presented how to show with SageMath that the, by Cui *et al.* so called, *improved* S-box S_{AIA} is in fact not even marginally an improvement.

Table 1 gives a survey of the MQ and the results of the algebraic attack resistance estimations scrutinized in this work.

We conclude, that in order to assess the resistance of an S-box against algebraic attacks it is not sufficient to derive some multivariate quadratic equation system and analyze it. Instead one would have to show that the derived MQ is optimal and superior to its Gröbner basis for solving and, thus, attacking it or the cipher it is used in.

S-box	MQ				Gröbner AIA/RD
	RD	RD	AIA/RD	AIA/RD	
maximal degree	2	2	2	2	7
# equations	23	16	2,039	32	8
# monomials	81	81	18,232	137	263
# dependent variables	8	8	2,032	8	8
$\log_2(\Gamma)$	22.9	27.2	24.0	52.0	–
$\log_2(\Gamma_{\text{CP}})$	13.4	20.0	28.5	20.5	–
SAT solver CPU time	0.6 s	0.7 s	7 s	0.8 s	0.15 s

Table 1: Survey of S-box MQ and estimations of their RAA.

References

- [1] Jie Cui, Hong Zhong, Jiankai Wang and Runhua Shi, *Generation and Optimization of Rijndael S-box Equation System*, Information Technology Journal, 13: 2482–2488, 2014.
- [2] William A. Stein *et al.*, *Sage Mathematics Software (Version 7.4)*. The Sage Development Team, 2016, <http://sagemath.org>.
- [3] Nicolas T. Courtois and Josef Pieprzyk, *Cryptanalysis of Block Ciphers with Overdefined Systems of Equations*, Proc. of Asiacrypt 2002, LNCS 2501, Springer-Verlag, 267–287, 2002.
An extended version of this paper is available on eprint.iacr.org <http://eprint.iacr.org/2002/044>.
- [4] Joan Daemen and Vincent Rijmen, *The Design of Rijndael: AES – the Advanced Encryption Standard*, Springer-Verlag, 2002, <http://jda.noekeon.org/>.
- [5] NIST, *Specification for the Advanced Encryption Standard (AES)*, Federal Information Processing Standards Publication 197, 2001, <http://csrc.nist.gov/publications/fips/fips197/fips-197.pdf>.
- [6] Wikipedia: SageMath, Sage (mathematics software), <https://en.wikipedia.org/wiki/SageMath>.
- [7] SageMathCloud, collaborative computational mathematics and course management; founder and main architect is William Stein, <https://cloud.sagemath.com/>.

- [8] The Sage code and its output for generating S-box multivariate quadratic polynomial equation systems at SageMathCloud: https://cloud.sagemath.com/projects/f5a9e4b0-3adb-4152-a9ea-e3ecfbcaedcd/files/Generating_S-Box_MQ.sagews.
- [9] Jung Hee Cheon and Dong Hoon Lee, *Resistance of S-Boxes against Algebraic Attacks*, Lecture Notes in Computer Science 3017, 83–93, 2004.
- [10] Jie Cui, Liusheng Huang, Hong Zhong, Chinchun Chang and Wei Yang, *An improved AES S-Box and its performance analysis*, Int. J. Innovative Comput. Inform. Control, 7: 2291–2302, 2011.
- [11] The Sage code and its output for calculating an S-box Lagrange Polynomial. SageMathCloud: <https://cloud.sagemath.com/projects/f5a9e4b0-3adb-4152-a9ea-e3ecfbcaedcd/files/S-BoxLagrangePolynomial.sagews>.
- [12] Michael Brickenstein and Alexander Dreyer, *PolyBoRi: A framework for Gröbner-basis computations with Boolean polynomials*, Journal of Symbolic Computation 44, 1326–1345, 2009, <http://dx.doi.org/10.1016/j.jsc.2008.02.017>.